
  Abstract--A technology template for the device 

simulation of a 3.3 kV Silicon Carbide (SiC) MOS  power 

transistor using the Crosslight NOVATCAD suite of 

simulation tools is provided.  The example illustrates the 

use of the process simulation tool CSUPREM to generate 

the basic mesh, epitaxial layer, and gate  structure. This 

structure is imported into the device simulator APSYS in 

which the doped layers are added using analytic 

approximations.  A methodology to simulate the SiC 

MOS electrical characteristics including the drain 

transfer curves, gate transfer curves, on-state resistance, 

and breakdown voltage is demonstrated. Simulation 

results are displayed and analyzed using the 

CrosslightView plotting tool. 

 

I.  INTRODUCTION 

 

Silicon carbide has become a leading candidate to 

replace silicon in high voltage power switching 

applications based on its superior material 

properties including high breakdown field, high 

thermal conductivity and high electron saturation 

velocity.  In order to properly understand and 

optimize these devices fast and accurate TCAD 

modeling is essential. This application note 

presents template files useful for the analysis, 

design and optimization of SiC MOSFET devices 

using the Crosslight NOVATCAD framework. 

CSUPREM is used to generate the simulation 

grid, epitaxial layer and gate structure.  This is 

imported into the APSYS device simulator where 

the doped regions are introduced analytically.  

Template files for simulating device Id-Vg, Id-Vd, 

and drain breakdown characteristics (BVdss) are 

provided.  

 

Fig. 1. shows schematic cross-section of the unit 

cell of a typical 3.3 kV SiC MOSFET [1].   The 

device is formed on a 30 m thick n-type 

epitaxial drift layer with a doping concentration 

of 3x1015 cm-3.  The channel length is assumed to 

be 1.0 m and the width of the JFET region is 1.5 

m.  The total cell pitch is 5.5 m.  The gate oxide 

is 50 nm thick while the gate contact is formed 

from n+ doped polysilicon. 

 

The simulation structure is shown in Fig. 2.  

Because of symmetry only the half-cell is 

simulated to minimize node count and run time.  

The p-well region is approximated as a two-step 

Gaussian profile.  The shallower profile sets the 

surface dopant concentration of approximately 

2x1017 cm-3 and thus determines the threshold 

voltage.  The deeper profile is more highly doped 

and serves to prevent punch-through from the 

drain-to-source under high-voltage operation.  

The total well depth is 1.0 m.  Shallower, higher 

concentration profiles form the n+ source and p+ 

body tie regions.   

 

 
Fig. 1 Schematic Cross Section of SiC 

MOSFET 

 

 
   

 

Fig. 2 Simulation Cross Section 

 

II. STRUCTURE GENERATION 

 

A. Initial set-up 

 

CSUPREM is used as a convenient method to 

define the material regions, generate the 

simulation grid and specify the epitaxial layer 

doping.  Minimal process simulation is actually 

performed since the well and source regions are 

defined analytically in the APSYS tool. The 

meshing strategy implements a coarse mesh deep 

in the structure with much finer spacings near the 

surface and at the edge of the p-well where the 

highest electric fields occur. This is easily 

accomplished using the RATIO command in the 

initial mesh setup.  This example employs 

approximately 5000 mesh points which provides 

a good compromise between required accuracy 

and sufficiently fast run time. The CSUPREM 

input file is provided in Appendix A. 

 
B. Physical model selection 

 

The most important physical models necessary 

for accurate simulation of SiC MOSFET devices 

are bulk mobility, inversion layer mobility, and 

impact ionization. These models have been 

discussed in detail in the literature [2-3].  

 

The bulk mobility model is based on a modified 

Arora formulation to capture both impurity and 

temperature dependence.  

 

𝜇𝑛 = 𝜇𝑚𝑖𝑛 +
𝜇𝑑

1+(
𝑁𝑡
𝑁0

)𝑎
                       (1) 

The electron mobility temperature dependence is 

modeled as 

𝜇(𝑇) = 𝜇300(
𝑇

300
)−2.4                      (2) 

In (1) d= 791cm2/V-sec, min= 33cm2/V-sec,  

N0= 1.94x1017cm-3, a=0.61 and T is the Kelvin 

temperature. The mobility coefficients are 

defined for the x-direction, which is parallel to the 

wafer surface. These are defined using the 

low_field_mobility statement in APSYS 

as illustrated below. 
 

low_field_mobility_model 

el_doping_dependence_model=arora 

&& 

el_ar_d=0.0791 el_ar_umin=0.0033 

el_ar_n=1.94E+23 el_ar_a=0.61 

 

The anisotropy in the low field mobility is 

captured as 

 

                              𝜇𝑦 = 1.2𝜇𝑥                                (3) 

 

The inversion layer mobility is SiC is determined 

by a number of scattering mechanisms the most 

important of which are the surface phonon 

scattering and the surface roughness scattering 

due to structural disprder at the SiC/SiO2 

interface.  These effects are captured in APSYS 

using the Lombardi mobility model as illustrated 

below. 

 
mobility_xy 

elec_field_model=lombardi 

hole_field_model=lombardi && 

elec_lombardi_b=1.00E+06  

elec_lombardi_alpha=1.746E+05 && 

elec_lombardi_beta=0.043 

elec_lombardi_delta=3.0E+14 && 

dir=y factor_elec=1.2 mater=1 

 

The model coefficients are calibrated to the 

measured data but are similar to those presented 

by Kimoto and Cooper [4]. 

 

Impact ionization calculations use Selberher’s 

formulation of the Chynoweth model [5] 

𝛼𝑛,𝑝 = 𝛼𝑛,𝑝
∞ 𝑒−(

𝐹𝑛

𝐹
)𝜅

                         (4) 

The ionization coefficients are taken from Loh et. 

al. [6] 

 

III. SiC SIMULATION IN NOVATCAD 

 
A. Forward I-V simulation 

 

The SiC MOSFET device structure generated 

with the CSUPREM export statement is the input 

structure for the device simulations. The 

load_mesh statement is used to enter the 

device structure into APSYS while the doping 

statements generate the P-well, source, body tie 

and JFET regions.  The subsequent 

suprem_to_apsys_material command 

maps the CSUPREM materials into the APSYS 

materials.  suprem_mater is the original 

material number in the CSUPREM file, which by 

default defines 4H-SiC as material 26. 

apsys_mater is the corresponding material 

number in APSYS.   The load_macro 

statements are used to describe the properties of 

the various materials used in the structure as 

defined in the materials database.  The 

suprem_contact statements are used to 



define the electrodes on the mesh imported from 

CSUPREM.    

 

The interface statement is used to specify the 

oxide charge, in units of 1/m2.  The location of the 

interface must correspond to the physical 

interface between the gate oxide and the SiC 

active layer.  The carrier mobilities are set by the 

low_field_mobility and mobility_xy 

statements described previously.  

 

Fig. 3 illustrates the simulated forward Id-Vd 

characteristics for the SiC MOSFET of Fig. 2. 

The Id-Vg characteristics are shown in Fig. 4. 

The data show a threshold voltage of 6.3 V.  The 

simulated on-state resistance at Vg=15 V is 14.5 

mohm-cm2 which is typical of commercial 3.3 kV 

rated SiC MOSFETs [1].  The simulation data 

include the effect of a lumped resistance equal to 

the actual substrate resistance.   

 
Fig. 3 Id-Vd Simulation Vgsmax=20V, 5V 

increments 

 

A complete listing of the APSYS input file for the 

Id-Vd simulations is given in appendix B.  

Appendix C presents the Id-Vg input file. 

 

b.) Reverse Simulations 

 

The APSYS input file for the reverse simulations 

is provided in Appendix D.  A key issue in the 

reverse simulation of wide-bandgap devices is the 

potential problem of convergence due to  

 
Fig. 4 Id-Vg Simulation Vds=0.1V,  

 

 

the very low equilibrium carrier densities.  The 

set_minority_carrier statement is used 

to artificially increase the minority carrier density 

thus alleviating this problem.  The I-V 

characteristics are computed using the scan 

statements.  Initially the drain voltage is swept 

until a specified avalanche current is reached as 

defined by the auto_finish and 

auto_until statements.  After this the 

simulation switches to a current boundary 

condition to allow the rest of the I-V curve 

computed. 

 

Accurate simulation of breakdown voltage 

requires proper choice of the impact ionization 

model and its associated parameters. 

NOVATCAD provides a number of different 

models commonly found in the literature to allow 

the user flexibility to choose the most appropriate 

model for a given simulation and to provide 

properly calibrated input parameters. The well-

known Chynoweth model with the appropriate 

input parameters described previously is 

specified by the impact_chynoweth 

statement. The simulated reverse I-V 

characteristics are shown in Fig. 5.  The simulated 

breakdown voltage is 4.0 kV in good agreement 

with the measured value from [1]. 

Fig. 5 BVdss Simulation 

Vgs=0.0V 



 

Appendix A: CSUPREM Input File 
 
 

############################################### 

# 

#  Grid structure  

# 

############################################### 

 

 

line x loc=    0.0     spacing=  0.1     tag=lft 

line x loc=    5.5     spacing=  0.1     tag=rht 

 

line y loc=    0.0     spacing=  0.5     tag=top 

line y loc=    5.0     spacing=  0.5     tag=bot 

           

 

eliminate y.direction xlo=0 xhi=10 ylo=1 yhi=5 ntimes=2 

option  auto.mesh.implant=f 

 

############################################### 

# 

#  Define structure as SiC  

# 

############################################### 

 

 

region  4H-SiC xlo=lft xhi=rht ylo=top yhi=bot 

bound exposed xlo=lft xhi=rht ylo=top yhi=top 

bound backside xlo=lft xhi=rht ylo=bot yhi=bot 

  

############################################### 

# 

#  Initialize substrate 

# 

############################################### 

 

init phosphorus conc=1.e18 

 

 

############################################### 

# 

#  SiC epi layer 

# 

############################################### 

 

deposit 4H-SiC thick=20 meshlayer=40 phosphorus conc=3e15  

deposit 4H-SiC thick=9.5 meshlayer=25 phosphorus conc=3e15 

ratio.layer=0.8  



deposit 4H-SiC thick=0.5 meshlayer=10 phosphorus conc=3e15 

diffuse time=1 temp=900 nitrogen 

 

 

############################################### 

# 

#  Deposit gate oxide   

# 

############################################### 

 

deposit oxide thick=0.05 divisions=5  

 

 

############################################### 

# 

#  Deposit and pattern gate   

# 

############################################### 

 

deposit poly thick=0.5 divisions=5 phosphorus conc=1E20  

etch poly left p1.x=1.5 

 

############################################### 

# 

#  Etch source region contacts   

# 

############################################### 

 

etch oxide left p1.x=1.5 

 

############################################### 

# 

#  Set activation and save final structure   

# 

############################################### 

 

activation.mode phosphorus fraction=1 force.activation=t 

activation.mode boron fraction=0.65 force.activation=t 

 

struct outf=Suprem_SIC_MOS.str 

 

export outf=SiC_MOS.aps xpsize=0.001 

 

 

quit 

 

 

 

 

 

 



 

Appendix B:  SiC MOSFET Id-Vd Simulation Input File 
 
begin 

 

$------- Polarity of Posititive Current Flow ----------- 

convention positive_current_flow=inward 

temperature temp=300 

 

load_mesh mesh_inf=SiC_MOS.aps  suprem_import=yes 

output sol_outf=MOS_Id_Vd.out 

more_output space_charge=yes 

 

$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$

$$ 

 

$  Define the doping profiles analytically 

 

$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$

$$ 

 

$ Well 

doping impurity=shal_dopant charge_type=acceptor max_conc=3e23  && 

  x_prof= ( 0.0, 4.0, 0.01, 0.01 )  && 

  y_prof= ( 29.0, 30.0, 0.0, 0.00 ) 

$ Retrograde 

doping impurity=shal_dopant charge_type=acceptor max_conc=8e23  && 

  x_prof= ( 0.0, 4.0, 0.01, 0.01 )  && 

  y_prof= ( 29.0, 29.5, 0.0, 0.00 ) 

$ Source 

doping impurity=shal_dopant charge_type=donor max_conc=1.e25  && 

  x_prof= ( 1.0, 3.0, 0.01, 0.01 )  && 

  y_prof= ( 29.8, 30.0, 0.0, 0.00 ) 

$ JFET 

doping impurity=shal_dopant charge_type=donor max_conc=1.e23  && 

  x_prof= ( 0.0, 5.5, 0.01, 0.01 )  && 

  y_prof= ( 29.1, 30.0, 0.0, 0.00 ) 

$Body tie 

doping impurity=shal_dopant charge_type=acceptor max_conc=1.e25  && 

  x_prof= ( 0.0, 0.8, 0.01, 0.01 )  && 

  y_prof= ( 29.8, 30.0, 0.0, 0.00 ) 

 

 

$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$

$$ 

 

$  Do the material mapping 

 

$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$

$$ 

 



 

suprem_property user_material_mapping = yes 

 

suprem_to_apsys_material suprem_mater=26  apsys_mater=1 

suprem_to_apsys_material suprem_mater=4 apsys_mater=3 

suprem_to_apsys_material suprem_mater=1 apsys_mater=2 

 

material_label_define label=4h_sic mater=   1 

load_macro name=4h_sic mater=   1 

 

material_label_define label=sio2 mater=   2 

load_macro name= sio2 mater= 2 

 

material_label_define label=poly mater=   3 

load_macro name= poly mater= 3 

 

 

 

$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$

$$ 

$ 

$ Define the contacts 

$ 

$ Drain/Substrate= 1 

$ Gate= 2 

$ Source= 3 

$ 

$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$

$$ 

 

suprem_contact num=1 xrange=(0.000000 10.000000)  yrange=(-9999.000000 

&& 

9999.000000) touch_mater=1 side=lower 

 

suprem_contact num=2 xrange=(2.5 10.000000)  yrange=(-9999.000000 && 

9999.000000) touch_mater=3 side=upper 

 

suprem_contact num=3 xrange=(0.0 1.4)  yrange=(-9999.000000 && 

9999.000000) touch_mater=1 side=upper 

 

contact num=1 

contact num=2 

contact num=3 

 

 

$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$

$$$ 

$ 

$  Set the bulk mobility parameters to accommodate mobility anisotropy 

$  Adjust low-field bulk mobilityfor a-axis =1/1.2 x c-axis values 

$  Get proper c-axis mobility from mobility anisotropy dir_y 

factor_elec=1.2 



$ 

$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$

$$ 

 

low_field_mobility_model el_doping_dependence_model=arora && 

el_ar_d=0.0791 el_ar_umin=0.0033 el_ar_n=1.94E+23 el_ar_a=0.61 

 

$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$

$$ 

 

 

$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$

$$ 

 

$  Set the channel mobility parameters 

$  Use optimized Lombardi coefficients based on Kimoto and Cooper 

 

$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$

$$ 

 

 

mobility_xy elec_field_model=lombardi hole_field_model=lombardi && 

elec_lombardi_b=1.00E+06  elec_lombardi_alpha=2.2E+05 && 

elec_lombardi_beta=0.043 elec_lombardi_delta=8.0E+14 && 

dir=y factor_elec=1.2 mater=1 

 

$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$

$$ 

 

$  Define the oxide fixed charge 

 

$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$

$$ 

 

 

interface model=charge within_y=(29.9, 30.1) within_x=(1.5, 5.5) && 

 fix_charge=1e14 

 

 

$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$

$$$ 

$ 

$   Equilibrium solution 

$ 

$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$

$$$$ 

 

newton_par damping_step=5. max_iter=150 print_flag=3 && 

var_tol=0.1 res_tol=0.1 

equilibrium 

 



$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$

$$$$ 

$ 

$   Rds solution--Set Vgs=15V to calibrate to 

$   published data for 3.3 kV MOSFET 

$ 

$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$

$$$$ 

 

 

newton_par damping_step=8.   && 

  var_tol=0.01 res_tol=0.01 change_variable=yes 

 

$------------------------------------------------------------- 

$ Vg=5.0 Trace 

$------------------------------------------------------------- 

 

scan var=voltage_2 value_to=5 min_step=1e-9 max_step=0.5 && 

init_step=0.001 

 

scan var=voltage_1 value_to=3 min_step=1e-6 max_step=0.2 && 

init_step=0.001 

 

scan var=voltage_1 value_to=0 min_step=1e-6 max_step=0.2 && 

init_step=0.001 

 

$------------------------------------------------------------- 

$ Vg=10.0 Trace 

$------------------------------------------------------------- 

 

scan var=voltage_2 value_to=10 min_step=1e-9 max_step=0.5 && 

init_step=0.001 

 

scan var=voltage_1 value_to=3 min_step=1e-6 max_step=0.2 && 

init_step=0.001 

 

scan var=voltage_1 value_to=0 min_step=1e-6 max_step=0.2 && 

init_step=0.001 

 

$------------------------------------------------------------- 

$ Vg=15.0 Trace 

$------------------------------------------------------------- 

 

scan var=voltage_2 value_to=15 min_step=1e-9 max_step=0.5 && 

init_step=0.001 

 

scan var=voltage_1 value_to=3 min_step=1e-6 max_step=0.2 && 

init_step=0.001 

 

scan var=voltage_1 value_to=0 min_step=1e-6 max_step=0.2 && 

init_step=0.001 

 



$------------------------------------------------------------- 

$ Vg=20.0 Trace 

$------------------------------------------------------------- 

 

scan var=voltage_2 value_to=20 min_step=1e-9 max_step=0.5 && 

init_step=0.001 

 

scan var=voltage_1 value_to=3 min_step=1e-6 max_step=0.2 && 

init_step=0.001 

 

end 

 

Appendix C: SiC MOSFET Id-Vg Simulation Input File 
 
begin 

 

$------- Polarity of Posititive Current Flow ----------- 

convention positive_current_flow=inward 

temperature temp=300 

 

load_mesh mesh_inf=SiC_MOS.aps  suprem_import=yes 

output sol_outf=MOS_Id_Vg.out 

more_output space_charge=yes 

 

$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$

$$$$$$$$$$$$$$$ 

 

$  Define the doping profiles analytically 

 

$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$

$$$$$$$$$$$$$$$ 

 

$ Well 

doping impurity=shal_dopant charge_type=acceptor max_conc=3e23  && 

  x_prof= ( 0.0, 4.0, 0.01, 0.01 )  && 

  y_prof= ( 29.0, 30.0, 0.0, 0.00 ) 

$ Retrograde 

doping impurity=shal_dopant charge_type=acceptor max_conc=8e23  && 

  x_prof= ( 0.0, 4.0, 0.01, 0.01 )  && 

  y_prof= ( 29.0, 29.5, 0.0, 0.00 ) 

$ Source 

doping impurity=shal_dopant charge_type=donor max_conc=1.e25  && 

  x_prof= ( 1.0, 3.0, 0.01, 0.01 )  && 

  y_prof= ( 29.8, 30.0, 0.0, 0.00 ) 

$ JFET 

doping impurity=shal_dopant charge_type=donor max_conc=1.e23  && 

  x_prof= ( 0.0, 5.5, 0.01, 0.01 )  && 

  y_prof= ( 29.1, 30.0, 0.0, 0.00 ) 

$Body tie 

doping impurity=shal_dopant charge_type=acceptor max_conc=1.e25  && 

  x_prof= ( 0.0, 0.8, 0.01, 0.01 )  && 



  y_prof= ( 29.8, 30.0, 0.0, 0.00 ) 

 

 

$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$

$$$$$$$$$$$$$$$ 

 

$  Do the material mapping 

 

$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$

$$$$$$$$$$$$$$$ 

 

 

suprem_property user_material_mapping = yes 

 

suprem_to_apsys_material suprem_mater=26  apsys_mater=1 

suprem_to_apsys_material suprem_mater=4 apsys_mater=3 

suprem_to_apsys_material suprem_mater=1 apsys_mater=2 

 

material_label_define label=4h_sic mater=   1 

load_macro name=4h_sic mater=   1 

 

material_label_define label=sio2 mater=   2 

load_macro name= sio2 mater= 2 

 

material_label_define label=poly mater=   3 

load_macro name= poly mater= 3 

 

 

 

$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$

$$ 

$ 

$ Define the contacts 

$ 

$ Drain/Substrate= 1 

$ Gate= 2 

$ Source= 3 

$ 

$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$

$$$ 

 

suprem_contact num=1 xrange=(0.000000 10.000000)  yrange=(-9999.000000 

&& 

9999.000000) touch_mater=1 side=lower 

 

suprem_contact num=2 xrange=(2.5 10.000000)  yrange=(-9999.000000 && 

9999.000000) touch_mater=3 side=upper 

 

suprem_contact num=3 xrange=(0.0 1.4)  yrange=(-9999.000000 && 

9999.000000) touch_mater=1 side=upper 

 

contact num=1 



contact num=2 

contact num=3 

 

$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$

$$ 

 

$  Set the bulk mobility parameters to accommodate mobility anisotropy 

$  Adjust low-field bulk mobilityfor a-axis =1/1.2 x c-axis values 

$  Get proper c-axis mobility from mobility anisotropy dir_y 

factor_elec=1.2 

 

$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$

$$ 

 

low_field_mobility_model el_doping_dependence_model=arora && 

el_ar_d=0.0791 el_ar_umin=0.0033 el_ar_n=1.94E+23 el_ar_a=0.61 

 

 

$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$

$$ 

 

 

$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$

$$ 

 

$  Set the channel mobility parameters 

$  Use optimized Lombardi coefficients based on Kimoto and Cooper 

 

$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$

$$ 

 

 

mobility_xy elec_field_model=lombardi hole_field_model=lombardi && 

elec_lombardi_b=1.00E+06  elec_lombardi_alpha=2.2E+05 && 

elec_lombardi_beta=0.043 elec_lombardi_delta=8.0E+14 && 

dir=y factor_elec=1.2 mater=1 

 

$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$

$$ 

 

$  Define the oxide fixed charge 

 

$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$

$$ 

 

 

interface model=charge within_y=(29.9, 30.1) within_x=(1.5, 5.5) && 

 fix_charge=1e14 

 

 

 



$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$

$$ 

$ 

$   Equilibrium solution 

$ 

$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$

$$ 

 

newton_par damping_step=5. max_iter=150 print_flag=3 && 

var_tol=0.1 res_tol=0.1 

equilibrium 

 

$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$

$$ 

$ 

$   Id-Vg solution--Set drain voltage to 0.1V 

$   and sweep Vg.   

$    

$ 

$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$

$$ 

 

 

newton_par damping_step=8.   && 

  var_tol=0.01 res_tol=0.01 change_variable=yes 

 

$Set the gate voltage 

scan var=voltage_1 value_to=0.1  && 

     init_step=0.01 max_step=1.0 min_step=1e-8 

 

$ Sweep the drain voltage 

scan var=voltage_2 value_to=15  && 

     init_step=0.01 max_step=0.25 min_step=1e-8 

 

 

end 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 



Appendix D: SiC MOSFET BVdss Simulation Input File 
 

 

 
begin 

 

$------- Part 1: Polarity of Potistive Current Flow ----------- 

convention positive_current_flow=inward 

temperature temp=300 

 

load_mesh mesh_inf=SiC_MOS.aps  suprem_import=yes 

output sol_outf=MOS_BV.out 

more_output impact_ionization=yes space_charge=yes 

 

$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$ 

 

$  Define well and source doping profiles analytically 

 

$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$ 

 

 

$ Well 

doping impurity=shal_dopant charge_type=acceptor max_conc=3e23  && 

  x_prof= ( 0.0, 4.0, 0.01, 0.01 )  && 

  y_prof= ( 29.0, 30.0, 0.0, 0.00 ) 

$ Retrograde 

doping impurity=shal_dopant charge_type=acceptor max_conc=8e23  && 

  x_prof= ( 0.0, 4.0, 0.01, 0.01 )  && 

  y_prof= ( 29.0, 29.5, 0.0, 0.00 ) 

$ Source 

doping impurity=shal_dopant charge_type=donor max_conc=1.e25  && 

  x_prof= ( 1.0, 3.0, 0.01, 0.01 )  && 

  y_prof= ( 29.8, 30.0, 0.0, 0.00 ) 

$ JFET 

doping impurity=shal_dopant charge_type=donor max_conc=1.e23  && 

  x_prof= ( 0.0, 5.5, 0.01, 0.01 )  && 

  y_prof= ( 29.1, 30.0, 0.0, 0.00 ) 

$Body tie 

doping impurity=shal_dopant charge_type=acceptor max_conc=1.e25  && 

  x_prof= ( 0.0, 1.0, 0.01, 0.01 )  && 

  y_prof= ( 29.8, 30.0, 0.0, 0.00 ) 

 

$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$ 

 

$  Do the material mapping 

 

$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$ 

 

 

suprem_property user_material_mapping = yes 

 



suprem_to_apsys_material suprem_mater=26  apsys_mater=1 

suprem_to_apsys_material suprem_mater=4 apsys_mater=3 

suprem_to_apsys_material suprem_mater=1 apsys_mater=2 

 

material_label_define label=4h_sic mater=   1 

load_macro name=4h_sic mater=   1 

 

material_label_define label=sio2 mater=   2 

load_macro name= sio2 mater= 2 

 

material_label_define label=poly mater=   3 

load_macro name= poly mater= 3 

 

 

 

$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$ 

$ 

$ Define the contacts 

$ 

$ Substrate= 1 

$ Gate= 2 

$ Source= 3 

$ 

$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$ 

 

suprem_contact num=1 xrange=(0.000000 10.000000)  yrange=(-9999.000000 

&& 

9999.000000) touch_mater=1 side=lower 

 

suprem_contact num=2 xrange=(2.5 10.000000)  yrange=(-9999.000000 && 

9999.000000) touch_mater=3 side=upper 

 

suprem_contact num=3 xrange=(0.0 1.4)  yrange=(-9999.000000 && 

9999.000000) touch_mater=1 side=upper 

 

contact num=1 

contact num=2 

contact num=3 

 

 

$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$ 

 

$  Set the bulk mobility parameters 

$  Adjust for a-axis =1/1.2 x c-axis values 

$  Get proper c-axis mobility from mobility anisotropy dir_y 

factor_elec=1.2 

 

$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$ 

 

low_field_mobility_model el_doping_dependence_model=arora && 

el_ar_d=0.0791 el_ar_umin=0.0033 el_ar_n=1.94E+23 el_ar_a=0.61 

 



 

$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$ 

 

 

$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$

$$ 

 

$  Set the channel mobility parameters 

$  Use optimized Lombardi coefficients based on Kimoto and Cooper 

 

$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$

$$ 

 

 

 

mobility_xy elec_field_model=lombardi hole_field_model=lombardi && 

elec_lombardi_b=1.00E+06  elec_lombardi_alpha=2.2E+05 && 

elec_lombardi_beta=0.043 elec_lombardi_delta=8.0E+14 && 

dir=y factor_elec=1.2 mater=1 

 

 

 

 

$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$

$$ 

$ 

$  Set virtual minority carrier concentration to improve convergence 

$  This is critical for wide-bandgap BV simulation 

$ 

$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$

$$ 

 

set_minority_carrier virtual_eg_kt=25 

 

 

 

$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$

$$ 

$ 

$ 

$   Impact ionization models for SiC 

$   Choose coefficients from Loh et al, TED 2008 

$   in this example 

$ 

$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$

$$ 

 

$  Originals from Crosslight Example 

$impact_chynoweth mater = 1 hole_setnum = 1  && 

$    hole_set1 =  [1.12e+09,1.72e+09,1.]  && 

$    elec_setnum = 1 elec_set1 =  [4.60e+07,1.78e+09,1.] 

 



$  Old Silvaco defaults 

$impact_chynoweth mater = 1 hole_setnum = 1  && 

$    hole_set1 =  [5.18e+08,1.4e+09,1.]  && 

$    elec_setnum = 1 elec_set1 =  [1.660e+08,1.27e+09,1.] 

 

$ Loh/GE Coefficinets 

impact_chynoweth mater = 1 hole_setnum = 1  && 

   hole_set1 =  [3.51e+08,1.03e+09,1.09]  && 

    elec_setnum = 1 elec_set1 =  [2.78e+08,1.05e+09,1.37] 

 

 

$ From Hatakeyama's paper  a axis coefficients 

$impact_chynoweth mater = 1 hole_setnum = 1  && 

$   hole_set1 =  [2.96E+09,1.6E+09,1.]  && 

$   elec_setnum = 1 elec_set1 =  [2.1e9,1.7E+09,1.] 

 

 

$ From Hatakeyama's paper  C axis coefficients 

$impact_chynoweth mater = 1 hole_setnum = 1  && 

$    hole_set1 =  [3.41e+10,2.5E+09,1.]  && 

$    elec_setnum = 1 elec_set1 =  [1.76e+10,3.3E+09,1.] 

 

 

$impact_lackner elec_a_coef=2.10E+09 elec_critical_field=1.7E+09 && 

$               hole_a_coef=2.96e+09 hole_critical_field=1.6E+09 

 

$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$ 

$ 

$  Equilibrium solution 

$ 

$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$ 

 

 

newton_par damping_step=5. max_iter=150 && 

var_tol=0.1 res_tol=0.1 

$ 

equilibrium 

$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$ 

$ 

$  Reverse Simulation 

$ 

$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$ 

 

 

newton_par damping_step=5.   && 

  var_tol=0.01 res_tol=0.01 change_variable=no 

 

 

 

scan var=voltage_1 value_to=5000 && 

   init_step=0.01 max_step=100 min_step=1e-8 && 

   auto_finish=current_1 auto_until=0.1 auto_condition=above 



 

 

scan var=current_1 value_to=10  && 

  init_step=1e-1 min_step=1.e-10 max_step=1.e4 

 

 

 

 

end 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 


